Daily MME Meta Analysis

Adapting a method recently developed by FDA to analyze a related opioid methods question, we used meta analytic techniques to test the impact of the four definitions in the real-world. The general set up is to compare opioid use in FL vs. CA across the 4 definitions of daily MME. We previously observed that Florida had higher unadjusted levels of opioid use, presumably an interaction with an older population and the enactment of clinical pain management legislation. We took two approaches, 1) treating daily MME as categorical by comparing the proportion of "high dose" users among opioid recipients, and 2) comparing means of daily MME between the states in a continuous manner, stratified by medicines used for acute versus chronic pain.

In [7]:
```
di "Stata MP"
version
di "Notebook generated on $S_DATE at $S_TIME ET"
```

```
Stata MP

version 16.0

Notebook generated on 26 May 2021 at 11:20:41 ET
```

Comparing "High Dose" patients in CA and FL

Input dataset from table of high dose patients (>90 daily MME) among adult outpatient opioid recipients identified using the PDMP of each state.

In [2]:
```
di "===== Proportion of high dose patients FL vs CA greater than 90 daily MME ====="
di "D1. Sum of days supply"
    csi  87295 87078   1398296  2343792
di "D2. On-therapy days"
    csi 136995 140822  1348596   2290048
di "D3. Defined observation window"
    csi 97346 86407  1388245   2344463
di "D4. Maximum daily dose"
    csi 211429 249471  1274162   2181399
```

```
===== Proportion of high dose patients FL vs CA greater than 90 daily MME =====

Definition 1


                 |   Exposed   Unexposed  |      Total
-----------------+------------------------+------------
           Cases |     87295       87078  |     174373
        Noncases |   1398296     2343792  |    3742088
-----------------+------------------------+------------
           Total |   1485591     2430870  |    3916461
                 |                        |
            Risk |  .0587611    .0358217  |   .0445231
                 |                        |
                 |      Point estimate    |    [95% Conf. Interval]
                 |------------------------+------------------------
 Risk difference |         .0229394       |    .0224949    .0233839
      Risk ratio |         1.640376       |    1.625414    1.655475
 Attr. frac. ex. |         .3903837       |    .3847723    .3959439
 Attr. frac. pop |         .1954347       |
                 +-------------------------------------------------
                               chi2(1) = 11405.78  Pr>chi2 = 0.0000

Definition 2


                 |   Exposed   Unexposed  |      Total
-----------------+------------------------+------------
           Cases |    136995      140822  |     277817
        Noncases |   1348596     2290048  |    3638644
-----------------+------------------------+------------
           Total |   1485591     2430870  |    3916461
                 |                        |
            Risk |  .0922158    .0579307  |   .0709357
                 |                        |
                 |      Point estimate    |    [95% Conf. Interval]
                 |------------------------+------------------------
 Risk difference |         .0342851       |    .0337349    .0348353
      Risk ratio |          1.59183       |    1.580486    1.603256
 Attr. frac. ex. |         .3717922       |    .3672831    .3762692
 Attr. frac. pop |         .1833353       |
                 +-------------------------------------------------
                               chi2(1) = 16446.29  Pr>chi2 = 0.0000

Definition 3


                 |   Exposed   Unexposed  |      Total
-----------------+------------------------+------------
           Cases |     97346       86407  |     183753
        Noncases |   1388245     2344463  |    3732708
-----------------+------------------------+------------
           Total |   1485591     2430870  |    3916461
                 |                        |
            Risk |  .0655268    .0355457  |   .0469181
                 |                        |
                 |      Point estimate    |    [95% Conf. Interval]
                 |------------------------+------------------------
 Risk difference |         .0299811       |    .0295201    .0304421
      Risk ratio |         1.843451       |    1.827062    1.859988
 Attr. frac. ex. |         .4575392       |    .4526731    .4623621
 Attr. frac. pop |         .2423885       |
                 +-------------------------------------------------
                               chi2(1) = 18534.92  Pr>chi2 = 0.0000

Definition 4


                 |   Exposed   Unexposed  |      Total
-----------------+------------------------+------------
           Cases |    211429      249471  |     460900
        Noncases |   1274162     2181399  |    3455561
-----------------+------------------------+------------
           Total |   1485591     2430870  |    3916461
                 |                        |
            Risk |  .1423198    .1026262  |   .1176828
                 |                        |
                 |      Point estimate    |    [95% Conf. Interval]
                 |------------------------+------------------------
 Risk difference |         .0396936       |    .0390145    .0403727
      Risk ratio |         1.386778       |    1.379279    1.394318
 Attr. frac. ex. |         .2789041       |    .2749835    .2828035
 Attr. frac. pop |         .1279419       |
                 +-------------------------------------------------
                               chi2(1) = 13991.68  Pr>chi2 = 0.0000
```

Scrape "Risk ratio" and CIs into new input dataset. Create log-transformed variables to meet normal distribution assumption of meta analytic statistics.

In [3]:
```
clear all
qui: input definition irr ll ul str31 label
1  1.640376      1.625414    1.655475     "D1. Sum of days supply"
2   1.59183     1.580486    1.603256    "D2. Accounting for overlap days"
3   1.843451    1.827062    1.859988   "D3. Defined observation window"
4  1.386778      1.379279    1.394318   "D4. Maximum daily dose"
end

gen lnirr=ln(irr)
gen lnll=ln(ll)
gen lnul=ln(ul)

qui: meta set lnirr lnll lnul, studylabel(label)
```

```
. gen lnirr=ln(irr)

. gen lnll=ln(ll)

. gen lnul=ln(ul)

. qui: meta set lnirr lnll lnul, studylabel(label)
```

Run meta analysis command using fixed effects model. Since there is no sampling variation, fixed effects is the preferred *a priori* specification.

In [4]:
```
meta summarize, fixed eform
```

```
  Effect-size label:  Effect Size
        Effect size:  lnirr
          Std. Err.:  _meta_se
        Study label:  label

Meta-analysis summary                                Number of studies =      4
Fixed-effects model                                  Heterogeneity:
Method: Inverse-variance                                       I2 (%) =   99.91
                                                                   H2 = 1085.83

--------------------------------------------------------------------------------
                       Study |        exp(ES)    [95% Conf. Interval]  % Weight
-----------------------------+--------------------------------------------------
      D1. Sum of days supply |          1.640       1.625       1.655     15.27
D2. Accounting for overlap~s |          1.592       1.580       1.603     25.06
D3. Defined observation wi~w |          1.843       1.827       1.860     16.07
      D4. Maximum daily dose |          1.387       1.379       1.394     43.60
-----------------------------+--------------------------------------------------
                  exp(theta) |          1.542       1.536       1.547
--------------------------------------------------------------------------------
Test of theta = 0: z = 237.00                             Prob > |z| = 0.0000
Test of homogeneity: Q = chi2(3) = 3257.49                  Prob > Q = 0.0000
```

For the sake of completeness, random effects models are also run, using the Sidik-Jonkman `random(sj)` estimator because tau is expected to be large Veroniki et al., with DerSimonian–Laird `random(dl)` as well separately for comparison, but fixed effects (above) is the more technically correct model specification.

In [5]:
```
meta summarize, random(sj) eform
```

```
  Effect-size label:  Effect Size
        Effect size:  lnirr
          Std. Err.:  _meta_se
        Study label:  label

Meta-analysis summary                                Number of studies =      4
Random-effects model                                 Heterogeneity:
Method: Sidik-Jonkman                                            tau2 =  0.0137
                                                               I2 (%) =   99.90
                                                                   H2 =  954.41

--------------------------------------------------------------------------------
                       Study |        exp(ES)    [95% Conf. Interval]  % Weight
-----------------------------+--------------------------------------------------
      D1. Sum of days supply |          1.640       1.625       1.655     24.99
D2. Accounting for overlap~s |          1.592       1.580       1.603     25.00
D3. Defined observation wi~w |          1.843       1.827       1.860     24.99
      D4. Maximum daily dose |          1.387       1.379       1.394     25.02
-----------------------------+--------------------------------------------------
                  exp(theta) |          1.607       1.433       1.803
--------------------------------------------------------------------------------
Test of theta = 0: z = 8.11                               Prob > |z| = 0.0000
Test of homogeneity: Q = chi2(3) = 3257.49                  Prob > Q = 0.0000
```

In [6]:
```
meta summarize, random(dl) eform
```

```
  Effect-size label:  Effect Size
        Effect size:  lnirr
          Std. Err.:  _meta_se
        Study label:  label

Meta-analysis summary                                Number of studies =      4
Random-effects model                                 Heterogeneity:
Method: DerSimonian-Laird                                       tau2 =  0.0156
                                                               I2 (%) =   99.91
                                                                   H2 = 1085.83

--------------------------------------------------------------------------------
                       Study |        exp(ES)    [95% Conf. Interval]  % Weight
-----------------------------+--------------------------------------------------
      D1. Sum of days supply |          1.640       1.625       1.655     24.99
D2. Accounting for overlap~s |          1.592       1.580       1.603     25.00
D3. Defined observation wi~w |          1.843       1.827       1.860     24.99
      D4. Maximum daily dose |          1.387       1.379       1.394     25.01
-----------------------------+--------------------------------------------------
                  exp(theta) |          1.607       1.422       1.816
--------------------------------------------------------------------------------
Test of theta = 0: z = 7.61                               Prob > |z| = 0.0000
Test of homogeneity: Q = chi2(3) = 3257.49                  Prob > Q = 0.0000
```

Results are similar, but SJ is preferred based on simulations in Veroniki et al. The fixed effects model over emphasizes precision (e.g., confuses it for more information) in D4 due to the higher number of high dose patients. Since there is no sampling variation

Interpretation

The proportion of "high dose" patients was consitently higher in Florida across all variants. However, the magnitude of the difference varied greatly: 84.3% (95% CI: 82.7%, 86.0%) for Definition 3 (defined observation window); 64.0% (95% CI: 62.5%, 65.5%) for Definition 1 (sum of days supply); 59.2% (95% CI: 58.0%, 60.3%) for Definition 2 (accounting for overlap days); and 38.7% (95% CI: 37.9%, 39.4%) for Definition 4 (maximum daily dose). Metrics confirmed very high heterogenity between the definitions, with I2 greater than 99% and H2 of 1086, supported by tests of hetereogenity chi2 of 3257 on 3 degrees of freedom (p<0.0001), and overall effect z=237, with 1 degree of freedom and p<0.0001.

Meta Analysis of Means by Type of Opioid

In this meta analysis we examine the impact of definitional variation on acute vs. chronic pain patients, measured by opioid formulation type. We stratified the sample into three sub-groups: 1) patients receiving on only immediate-release or short-acting opioids labeled for acute pain (hereafter immediate-release; 2) patients receiving only extended-release or long-acting opioids generally labeled for chronic pain (hereafter extended-release); and 3) patients receiving both immediate-release and extended-release opioids contemporaneously within the 3 month observation period (e.g., chronic pain patients receiving opioids for breakthrough pain or during taper).

Continuing with the approach in the previous meta analysis, we calculated mean differences in daily MME between Florida and California, treating each of the 4 daily MME definitions as separate studies run on the same sample (e.g., fixed effects).

Immediate-release only

In [6]:
```
clear
input definition n_fl m_fl sd_fl n_ca m_ca sd_ca
1 1338828 34.0531498 28.4797412 2273028 30.3156249 222.6063485
2 1338828 35.0964146 30.180772 2273028 31.5819604 223.0198312
3 1338828 12.5794512 25.2892396 2273028 10.3398905 42.5422362
4 1338828 44.7478467 48.3917948 2273028 39.6430507 280.3601706
end

qui: meta esize n_fl m_fl sd_fl n_ca m_ca sd_ca, esize(mdiff)
meta summarize, fixed
```

```
     definit~n       n_fl       m_fl      sd_fl       n_ca       m_ca      sd_ca


  Effect-size label:  Mean Diff.
        Effect size:  _meta_es
          Std. Err.:  _meta_se

Meta-analysis summary                     Number of studies =      4
Fixed-effects model                       Heterogeneity:
Method: Inverse-variance                            I2 (%) =   98.63
                                                        H2 =   72.98

--------------------------------------------------------------------
            Study |     Mean Diff.    [95% Conf. Interval]  % Weight
------------------+-------------------------------------------------
          Study 1 |          3.738       3.359       4.116      3.92
          Study 2 |          3.514       3.135       3.894      3.90
          Study 3 |          2.240       2.160       2.319     89.72
          Study 4 |          5.105       4.626       5.584      2.45
------------------+-------------------------------------------------
            theta |          2.418       2.343       2.493
--------------------------------------------------------------------
Test of theta = 0: z = 63.18                   Prob > |z| = 0.0000
Test of homogeneity: Q = chi2(3) = 218.94        Prob > Q = 0.0000
```

Extended-release only

In [7]:
```
clear
input definition n_fl m_fl sd_fl n_ca m_ca sd_ca
1 26039 86.9071545 87.9504585 40038 90.2232825 100.0878302
2 26039 96.9302372 102.8249551 40038 103.7573329 134.372793
3 26039 66.8367252 81.142005 40038 72.753132 104.6161615
4 26039 143.0437107 159.4875273 40038 153.6802569 205.2125971
end

qui: meta esize n_fl m_fl sd_fl n_ca m_ca sd_ca, esize(mdiff)
meta summarize, fixed
```

```
     definit~n       n_fl       m_fl      sd_fl       n_ca       m_ca      sd_ca


  Effect-size label:  Mean Diff.
        Effect size:  _meta_es
          Std. Err.:  _meta_se

Meta-analysis summary                     Number of studies =      4
Fixed-effects model                       Heterogeneity:
Method: Inverse-variance                            I2 (%) =   86.38
                                                        H2 =    7.34

--------------------------------------------------------------------
            Study |     Mean Diff.    [95% Conf. Interval]  % Weight
------------------+-------------------------------------------------
          Study 1 |         -3.316      -4.806      -1.826     35.11
          Study 2 |         -6.827      -8.745      -4.909     21.19
          Study 3 |         -5.916      -7.415      -4.418     34.70
          Study 4 |        -10.637     -13.578      -7.695      9.01
------------------+-------------------------------------------------
            theta |         -5.622      -6.504      -4.739
--------------------------------------------------------------------
Test of theta = 0: z = -12.48                  Prob > |z| = 0.0000
Test of homogeneity: Q = chi2(3) = 22.03         Prob > Q = 0.0001
```

Both Extended-release and Immediate-release

In [8]:
```
clear
input definition n_fl m_fl sd_fl n_ca m_ca sd_ca
1 120724 82.95423 59.1676551 117804 74.1906194 64.4024217
2 120724 160.1525421 131.6299812 117804 143.9839494 151.4652358
3 120724 133.0969773 125.945819 117804 122.7372442 148.5490438
4 120724 267.949697 238.0130378 117804 250.7462218 282.0999741
end

qui: meta esize n_fl m_fl sd_fl n_ca m_ca sd_ca, esize(mdiff)
meta summarize, fixed
```

```
     definit~n       n_fl       m_fl      sd_fl       n_ca       m_ca      sd_ca


  Effect-size label:  Mean Diff.
        Effect size:  _meta_es
          Std. Err.:  _meta_se

Meta-analysis summary                     Number of studies =      4
Fixed-effects model                       Heterogeneity:
Method: Inverse-variance                            I2 (%) =   98.34
                                                        H2 =   60.27

--------------------------------------------------------------------
            Study |     Mean Diff.    [95% Conf. Interval]  % Weight
------------------+-------------------------------------------------
          Study 1 |          8.764       8.267       9.260     69.06
          Study 2 |         16.169      15.031      17.307     13.13
          Study 3 |         10.360       9.255      11.464     13.94
          Study 4 |         17.203      15.111      19.296      3.88
------------------+-------------------------------------------------
            theta |         10.286       9.873      10.698
--------------------------------------------------------------------
Test of theta = 0: z = 48.90                   Prob > |z| = 0.0000
Test of homogeneity: Q = chi2(3) = 180.81        Prob > Q = 0.0000
```

Interpretation

- ER only group had *lower* mean daily MME in Florida than California?!
- Heterogeneity by I^2 was high for all 3 definitions
- Heterogeneity was lowest for ER-only group by both I^2 and X^2
- For ER+IR group, the definitional variants would have resulted in us concluding that the average dose was 8.8 (8.3, 9.3) milligrams to 17.2 (15.1, 19.3) milligrams higher in Florida.